Escola Politécnica da Universidade de São Paulo

PME 3380 – Modelagem de Sistemas Dinâmicos

Lista E

Gabriela Vasconcelos Araujo - 10771497

Prof. Dr. Décio Crisol Donha
Prof. Dr. Agenor de Toledo Fleury
São Paulo, 2020

SUMÁRIO

EXERCÍCIO – EQUACIONAMENTO E FUNÇÃO DE TRANSFERÊNCIA

Nesta lista, busca-se trabalhar o sistema massa-mola-amortecedor abaixo exposto. Para tal, utiliza-se como coordenada generalizada x, referente a deformação da mola.

Figura 1: Sistema massa-mola-amortecedor



Aplicando a Segunda Lei de Newton, encontra-se a equação diferencial que rege o sistema estudado:

$$\ddot{x} = -\frac{k}{m}x - \frac{c}{m}\dot{x} + \frac{F(t)}{m}$$

Define-se, então, o seguinte espaço de estados:

$$\begin{bmatrix} x_1 \\ x_2 \end{bmatrix} = \begin{bmatrix} x \\ \dot{x} \end{bmatrix}; \begin{bmatrix} \dot{x_1} \\ \dot{x_2} \end{bmatrix} = \begin{bmatrix} \dot{x} \\ \ddot{x} \end{bmatrix}$$

$$\begin{bmatrix} \dot{x} \\ \ddot{x} \end{bmatrix} = \begin{bmatrix} 0 & 1 \\ -{k}/{m} & -{c}/{m} \end{bmatrix} \begin{bmatrix} x \\ \dot{x} \end{bmatrix} + \begin{bmatrix} 0 \\ 1 \end{bmatrix} F(t)$$

A fim de encontrar a função de transferência, aplica-se a transformada de Laplace:

$$\dot{x_1} = x_2 \xrightarrow{\mathcal{L}} sX_1 - x_1(0) = X_2 \ (1)$$

$$\dot{x_2} = -\frac{k}{m}x_1 - \frac{c}{m}x_2 + \frac{F(t)}{m} \xrightarrow{\mathcal{L}} sX_2 - x_2(0) = -\frac{k}{m}X_1 - \frac{c}{m}X_2 + \frac{F}{m} \ (2)$$

Isolando X_2 em (2) e substituindo em (1):

$$sX_1 - x_1(0) = \left(\frac{1}{ms + c}\right)\left(\frac{x_2(0)}{m} + kx_1 + F\right)$$

Como $x_1(0) = x(0) = 0$ e $x_2(0) = \dot{x}(0) = 0$:

$$X_1 = \frac{F}{ms^2 + cs + k}$$

Como $G(s) = {Y(s)}/{F(s)}$, considerando a saída $Y(s)$ como $X_1(s)$, chega-se na função de transferência:

$$G(s) = \frac{1}{ms^2 + cs + k}$$

CÓDIGO

A simulação abaixo exposta foi desenvolvida com os mesmos parâmetros do exemplo presente no enunciado ($m = 1\ kg$ e $k = 900\ N/m$). No código, o input 'zeta' se refere ao valor de ζ.

```
clear all

// Definição dos Parâmetros
m = 1; // [m] = kg
k = 900; // [k] = N/m
zeta = input('Zeta ='); // testar com os três casos (zeta < 1; zeta = 1; zeta > 1)
c = 2 * zeta * sqrt(k/m); // [c] = Ns/m

// Definição do Sistema Linear Usando o Comando "syslin"
A = [0 1; (-k/m) (-c/m)];
B = [0; 1/m];
C = [0 0];
D = [0];
MassaMolaAmortecedor = syslin('c', A, B, C, D);

// Definição do Vetor de Tempo
t = 0:0.01:2;
// Definição da Condição Inicial
x0 = [0; 0];
// Definição da Entrada
u = ones (2*t);
// Realização da Simulação com o Comando "csim"
[y,x] = csim(u, t, MassaMolaAmortecedor, x0);

f1 = scf(1);
plot(t,x);
h1 = legend(['x','xp']);
xtitle('Resposta do Sistema','Tempo (s)','Deformação da Mola (m) ou Velocidade do Bloco (m/s)');
xgrid;
```

SAÍDAS

O enunciado solicita simulações para três cenários:

- $\zeta = \frac{c}{2\sqrt{km}} < 1 \Rightarrow \zeta = 0{,}2$
- $\zeta = \frac{c}{2\sqrt{km}} = 1$
- $\zeta = \frac{c}{2\sqrt{km}} > 1 \Rightarrow \zeta = 10$

Figura 2: Resposta do Sistema para ζ = 0,2



Figura 3: Resposta do Sistema para ζ = 1



Figura 4: Resposta do Sistema para ζ = 10



LIÇÃO 1 – AUTOVALORES DE A E RAÍZES DE G(S)

A fim de se calcular os autovalores da matriz A, realizamos a seguinte operação:

$$\det(A - \lambda I) = 0$$

$$det \begin{bmatrix} -\lambda & 1 \\ -{}^{k}/_{m} & -{}^{c}/_{m} - \lambda \end{bmatrix} = 0$$

$$\lambda({}^{c}/_{m} + \lambda) + {}^{k}/_{m} = 0 \implies \lambda^2 + \frac{c}{m}\lambda + \frac{k}{m} = 0$$

Nota-se que as raízes do polinômio acima exposto são as mesmas do polinômio $G(s)$ para as condições iniciais dadas. Considerando $m = 1\ kg, k = 900\ N/m$ e $\zeta = 0{,}1$:

$$\lambda = \frac{-\frac{c}{m} \pm \sqrt{\left(\frac{c}{m}\right)^2 - 4\left(\frac{k}{m}\right)}}{2} \implies \begin{cases} \lambda_1 = -3 + 9\sqrt{11}i \\ \lambda_2 = -3 - 9\sqrt{11}i \end{cases}$$

É possível verificar, então, que o módulo dos autovalores é igual a frequência natural do sistema massa-mola-amortecedor:

$$\omega = \sqrt{\frac{k}{m}} = \sqrt{900} = 30\ rad/s$$

$$|\lambda_1| = |\lambda_2| = \sqrt{3^2 + \left(9\sqrt{11}\right)^2} = 30$$

$$\therefore \omega = |\lambda_1| = |\lambda_2| = 30$$

Nota-se também que, dividindo o módulo da parte real do número complexo pelo módulo do número complexo, obtém-se o coeficiente de amortecimento:

$$\frac{|\mathbb{R}(\lambda_1)|}{|\lambda_1|} = \frac{|\mathbb{R}(\lambda_2)|}{|\lambda_2|} = \frac{3}{30} = 0{,}1 = \zeta$$

Observa-se, ainda, que a frequência de oscilação é igual ao módulo da parte imaginária do polo:

$$|\mathbb{I}(\lambda_1)| = |\mathbb{I}(\lambda_2)| = 9\sqrt{11} = 29{,}85\ rad/s = \omega\sqrt{1 - \zeta^2}$$

LIÇÃO 2 – SIMULAÇÃO PARA CENÁRIOS DIVERSOS

Para esta tarefa, pede-se a simulação do exercício em caso de entrada nula para diferentes condições iniciais não nulas. Optou-se por utilizar as seguintes condições iniciais:

- $x(0) = [-3, -2.5, -2, -1.5, -1, 1, 1.5, 2, 2.5, 3]$
- $\dot{x}(0) = [-3, -2.5, -2, -1.5, -1, 1, 1.5, 2, 2.5, 3]$

CÓDIGO

```
clear all

// Definição dos Parâmetros
m = 1; // [m] = kg
k = 900; // [k] = N/m
zeta = input('Zeta ='); // testar com os três casos (zeta < 1; zeta = 1; zeta > 1)
c = 2 * zeta * sqrt(k/m); // [c] = Ns/m

// Definição do Vetor de Tempo
t = 0:0.01:2;

// Definição da Condição Inicial
x0 = [-3 -2.5 -2 -1.5 -1 1 1.5 2 2.5 3];
xp0 = [-3 -2.5 -2 -1.5 -1 1 1.5 2 2.5 3];

// Integração Numérica
funcprot(0)
  function dy=MassaMolaAmortecedor(t, y)
    dy(1) = y(2);
    dy(2) = -(k/m)*y(1) - (c/m)*y(2);
  endfunction
for i = 1:10
  y = ode([x0(i);xp0(i)],0,t,MassaMolaAmortecedor);
  for j=1:length(t)
    x(i,j) = y(1,j);
    xp(i,j) = y(2,j);
  end
end

//Plotagem
scf(1);
xtitle("Espaço de Fases");
xlabel("Deformação da Mola (m)");
ylabel(" Velocidade do Bloco (m/s)");
for i = 1:length(x0)
  plot(x(i,:),xp(i,:),'r');
end
xgrid
```

SAÍDAS

São obtidos, assim, os espaços de fases para três cenários distintos. Para o primeiro, em que se tem polos complexos ($\zeta < 1$), temos a Figura 5. Já para o caso de polos reais e iguais ($\zeta = 1$), temos a Figura 6. Por fim, para polos reais e distintos ($\zeta > 1$), temos a Figura 7.

Figura 5: Espaço de Fases para ζ = 0,2



Figura 6: Espaço de Fases para ζ = 1



Figura 7: Espaço de Fases para ζ = 10